DINAMICA DA BOLA SALTANDO

(BOUNCING BALL)

Modelagem e Simulação

Prof. Dr. Roberto Spinola Barbosa

São Paulo

1. Introdução

Os sistemas mecânicos podem estar submetidos a impactos durante a sua movimentação. Neste caso haverá variação repentina da velocidade sem mudança significativa de posição. Da segunda lei de *Newton* aplicada a uma partícula sabemos que $m\,\vec{a} = \vec{F}$ ou em diferencial:

$$m\frac{d\vec{V}}{dt} = \vec{F} \qquad \text{ou} \qquad m\,d\vec{V} = \vec{F}\,dt \tag{1}$$

expressão que integrada durante um intervalo infinitesimal de tempo ($t_1 \rightarrow t_2$) correspondente ao impacto, resulta em:

$$m\int_{V_1}^{V_2} d\vec{V} = \int_{t_1}^{t_2} \vec{F}\,dt \qquad \text{ou} \qquad m(\vec{V}_2 - \vec{V}_1) = \int_{t_1}^{t_2} \vec{F}\,dt \tag{2}$$

Chamando a integral definida $\vec{I} = \int_{t_1}^{t_2} \vec{F}\,dt$ de IMPULSO obtêm-se a expressão da variação da velocidade da partícula no impacto:

$$m\,\Delta\vec{V} = \vec{I} \tag{3}$$

A propriedade do centro de massa permite expressar o Teorema da Resultante dos Impulsos (*TRI*) para um corpo rígido e utilizando o momento da quantidade de movimento descreve-se o Teorema do Momento dos Impulsos (*TMI*) respectivamente como:

$$m\,\Delta\vec{V}_G = \vec{I}^{\,ext} \qquad \text{e} \qquad J_O\,\Delta\vec{\omega} = \vec{M}_O^{\,I} \tag{4}$$

Para impacto é utilizado a expressão empírica de “Coeficiente de Restituição” que relaciona a velocidade normal de aproximação entre os corpos (V^-) no instante imediatamente anterior ao impacto (t^-) e velocidade de afastamento (V^+) imediatamente posterior, descrita como:

$$V^+ = -e\,V^- \quad \text{onde} \quad V^+ = \left(\vec{V}_1^+ - \vec{V}_2^+\right)\cdot\vec{n} \quad \text{e} \quad V^- = \left(\vec{V}_1^- - \vec{V}_2^-\right)\cdot\vec{n} \tag{5}$$

onde o escalar ***e*** (podendo variar entre 0 e 1) caracteriza o impacto totalmente plástico ou elástico (admite-se que exista uma única normal entre as superfícies em contato). Está é a hipótese atribuída a *Newton*. Para mais informações, consulte o livro: Mecânica Geral (*França* e *Matsumura*, 2011).

2. Bola Saltando (*bouncing ball*)

Considere que uma bola é lançada com condições iniciais conhecidas. Determinar qual a trajetória da bola no espaço e sua interação com o solo, considerando a bola como um corpo rígido com 3 graus de liberdade e movimento no plano ***Oxy***, conforme mostrado na Figura 1. Sabe-se ainda que o contato com o solo durante o impacto é parcialmente elástico e sem escorregamento longitudinal.



Figura 1 – Bola em Movimento (plano Oxy)

Trata-se de um problema não linear com comportamento variável em função da posição. Será utilizada para a solução do problema a técnica de integração numérica determinando a dinâmica anterior ao impacto, com mudança das condições inicias no impacto e dinâmica posterior.

2.1. Modelo da dinâmica

O modelo do sistema "bola em queda livre" é obtido pela aplicação dos teoremas de dinâmica ao corpo livre. Para tanto o diagrama de forças sobre o corpo livre, foi elaborado indicando as ações de forças externas e impulsos no instante do impacto, conforme apresentado na Figura 2.



Figura 2 – Diagrama de forças sobre o corpo livre (DFCL)

Movimento de Translação

Os movimentos de translação são obtidos pela aplicação do Teorema da Resultante (*TR*) ao centro de massa do corpo:

$$m\frac{d^2(G-O)}{dt^2} = \sum \vec{F} \tag{6}$$

sendo que o lado direito da equação é obtido do diagrama de forças sobre o corpo livre da Figura 2

Movimento de Rotação

O movimento de rotação é obtido pela aplicação do Teorema da Quantidade de Movimento Angular (*TQMA*) tomando como pólo o centro de massa do corpo apenas da direção ortogonal ao plano ***Oxy***:

$$\frac{d}{dt}\left[\{\vec{i} \quad \vec{j} \quad \vec{k}\}[J]_O\{\omega\}\right] + (G-O)\wedge\vec{a}_O = \sum M_O^{ext} \qquad J_{zO}\alpha_z = \sum M_{zO}^{ext} \tag{7}$$

Equações de Movimento

Separando as equações do movimento plano em cada direção, obtêm-se:

$$\begin{Bmatrix} m\ddot{x} = 0 \\ m\ddot{y} = -mg \\ J_{zO}\dot{\omega} = 0 \end{Bmatrix} \tag{8}$$

2.2. Modelo do Impacto

O modelo do sistema "bola em impacto" é obtido pela aplicação dos teoremas de impacto. Para tanto o diagrama de velocidade do corpo livre foi elaborado, indicando as velocidades translacionais e angular imediatamente antes e depois do impacto parcialmente elástico sem escorregamento, conforme apresentado na Figura 3.



Figura 3 – Diagrama de Velocidades

Utilizando as expressões do *TRI* e *TMI* para o pólo ***descrito anteriormente***:

$$m\,\Delta\vec{V}_G = \vec{I}^{\,ext} \qquad \text{e} \qquad J_G\,\Delta\vec{\omega} = \vec{M}^{\,I}_G \tag{9}$$

e a relação do coeficiente de restituição $V^+ = -e\,V^-$ aplicada apenas na direção ***y*** e utilizando o diagrama de impulsos da Figura 2, obtêm-se:

$$\left\{\begin{matrix} m(V^+_{Gx} - V^-_{Gx}) = I_x \\ V^+_{Gy} = -e\,V^-_{Gy} \\ J_{zG}\left(\omega^+ - \omega^-\right) = r\,I_x \end{matrix}\right\} \tag{10}$$

Como o impacto ocorre sem escorregamento a velocidade na direção longitudinal no ponto ***C*** de contato durante o impacto permanece imóvel ou seja, $V^-_C = 0$ e $V^+_C = 0$ obtêm-se da formula de campo de velocidades no impacto:

$$\vec{V}_G = \vec{V}_C + \vec{\omega} \wedge (G - C) \qquad \text{ou} \qquad V_{Gx} = r\,\omega \tag{11}$$

Desta forma considerando a bola de massa ***m*** como uma esfera de raio ***r*** e momento de inércia $J_G = 2mr^2/5$ e resolvendo o sistema de equações algébricas, obtêm-se:

$$\begin{Bmatrix} V_{Gx}^{+} = \left(2r\omega^{-} + 5V_{Gx}^{-}\right)/7 \\ V_{Gy}^{+} = -eV_{Gy}^{-} \\ \omega^{+} = \left(2r\omega^{-} + 5V_{Gx}^{-}\right)/7r \end{Bmatrix} \tag{12}$$

Estas são as equações que relacionam as velocidade translacional (V^{+}) e rotacional (ω^{+}) no instante imediatamente posterior ao impacto com as velocidades anteriores ao impacto (V^{-} e ω^{-} respectivamente).

3. Simulação

Para a simulação do sistema não linear será utilizado um algoritmo de integração numérica. A simulação ocorre em três etapas sucessivas e repetidas:

- Dinâmica livre anterior ao impacto;
- Impacto;
- Dinâmica posterior (até o próximo impacto).

3.1. Implementação numérica

Para a implementação numérica será utilizado um vetor de espaço de estados para redução das equações diferenciais de segunda ordem para um sistema duplo de primeira ordem. O vetor de estado será formado pelas coordenadas independentes do sistema (conforme mostrada na Figura 1) e suas derivadas:

$$z = \{x \quad y \quad \theta \quad \dot{x} \quad \dot{y} \quad \omega\} \tag{13}$$

O sistema dinâmico fica descrito pelo conjunto de equações diferenciais de primeira ordem:

$$\{\dot{z}\} = [A]\{z\} + [B]\{u\} \tag{14}$$

onde $\dot{z} = \{\dot{x} \quad \dot{y} \quad \omega \quad \ddot{x} \quad \ddot{y} \quad \dot{\omega}\}$ e $\{u\}$ é o vetor de ações externas atuantes nos graus de liberdade segundo a proporção [B].

Devido a simplicidade do sistema, as matrizes [A] ; [B] e $\{u\}$ resultam em:

$$[A] = \begin{bmatrix} 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad ; \quad [B] = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ -1 \\ 0 \end{bmatrix} \quad \text{e} \quad \{u\} = g \tag{15}$$

O vetor com a derivada dos estados $\{\dot{z}\}$ deverá ser integrada numericamente utilizando o algoritmo *ODE* (*ordinary differential equation*) do programa *Scilab* (ou *Matlab*) para as condições iniciais $z_0 = \{x_0 \quad y_0 \quad \theta_0 \quad \dot{x}_0 \quad \dot{y}_0 \quad \omega_0\}$ até que seja atingida a cota y = 0 que corresponde a posição do impacto.

Considerando que os instantes de tempo para a saída ***z*** iniciam-se em ***t***0 no espaçamento ***t***, a sintaxe da instrução de **INTEGRAÇÃO NUMÉRICA** utilizando a função ODE será:

[z,rd] = ode("roots", z0, t0, t, bola, 1, interrup);

A função ***bola*** recebe os estados iniciais ***z***0 e o instante de tempo inicial ***t***0 e contem as equações de movimento conforme descrito acima:

function zpto = bola (t,z)

O processo se repete para os demais instantes de tempo ***t*** sempre utilizando o estado do instante anterior ***z*** até que a interrupção no instante de impacto seja identificada pela opção *"roots"* do integrador.

3.2. Tratamento do Impacto

No instante do impacto a integração deve ser interrompida, o vetor de estados deve ser substituído pelas condições inicias correspondentes aos estados posterior ao impacto e reiniciada.

$$z_0^+ = \{x^- \quad y^- \quad \theta^- \quad \dot{x}^+ \quad \dot{y}^+ \quad \omega^+\} \tag{16}$$

como por definição o impacto ocorre sem mudança substancial de sua posição, apenas as velocidades posteriores ao impacto são atualizadas, conforme a solução do problema de impacto descrito no item 2.2.

Para a interrupção do processo de integração numérica utilize-se de uma função auxiliar que verifica a interceptação do zero (*root finder*). Esta opção é selecionada dentro da chamada da função *ODE* ("*roots"*) que fornece além da saída ***z*** o instante ***rd*** do zero da função de interrupção.
A função de interrupção pode ser expressa em função do tempo ou de qualquer estado. Neste caso utiliza-se da coordenada ***y*** = 0 (que corresponde ao estado *z(2)*) como função para a interrupção.

A sintaxe do comando que faz a verificação da raiz da função é:

```
function w = interrup(t,z)      w =  z(2) ;      endfunction
```

4. Resultados

Como exemplo de verificação do desempenho deste algoritmo o movimento da **Bola Saltando** foi simulado. A bola foi lançada com velocidade inicial e velocidade angular e percorre uma trajetória elíptica livre até tocar o solo (altura igual a zero). Neste instante o processo de integração é interrompido, o problema de impacto é solucionado obtendo as velocidade translacional (V^+) e rotacional (ω^+) no instante imediatamente posterior ao impacto. Estes valores tornam-se as novas condições iniciais para a continuação da integração do movimento. Os resultados estão apresentados na Figura 4 para as condições iniciais descritas no anexo.



Figura 4 - Bola Saltando

5. Referencias Bibliográfica

França, L. N. F. Matsumura, A. Z. (2011) Mecânica Geral, Blucher, 3 edição. pp. 316.

Johnson, K. L. (1985) Contact Mechanics. Ed. Cambridge, pp. 452. ISBN-13: 978-0521347969

Stronge, W. J. (2000) Impact mechanics. Cambridge University Press 2000 New York. xix, pp. 280. ISBN-13: 978-0521602891.

Scilab (2018) Reference Manual, Scilab 6.0.1. https://www.scilab.org/en/download/6.0.1

6. ANEXO - A

Descrição do sistema dinâmico e condições iniciais para a simulação

// Dados do sistema:

- m = 0.623; // massa da bola em kg;
- p = 0.749; // perímetro da bola em metros;
- r = p/(2*%pi); // raio da bola em metros;
- Jg = 2*m*r*r/5; // momento de inércia da esfera;
- g = 9.81; // Aceleração local da gravidade em m/s^2;
- ey = -0.8; // Coeficiente de restituição na direção y;
- Fy = -g;

// Atribuição das condições iniciais:

- x0 = 0.0; // Posição longitudinal inicial da bola;
- y0 = 2.0; // Posição vertical inicial da bola;
- omega0 = 0.0; // Posição angular inicial da bola;
- xpto0 = 3.0; // Velocidade longitudinal inicial da bola;
- ypto0 = 6.0; // Velocidade vertical inicial da bola;
- omegapto0 = -90.0; // Velocidade angular inicial da bola

7. ANEXO - B

Códigos do Programa de Integração Numérica no programa SCILAB

```
// CODIGO PARA SIMULAR BOLA SALTANDO
// SCILAB 6.0.0
// Copyright 2018 Roberto Spinola Barbosa
// Versão 06 de junho de 2017
//

clc                       // limpa a tela
clear                     // apaga todas as variáveis da memória
xdel(winsid())            // fecha todas as figuras

//Dados do sistema:
m = 0.623;                // massa da bola em kg
p = 0.749;                // perímetro da bola em metros
r = p/(2*%pi);            // raio da bola em metros
Jg = 2*m*r*r/5;           // momento de inércia da esfera
g = 9.81;                 // Aceleração local da gravidade em m/s^2
ey = -0.8;                // Coeficiente de restituição na direção y

// Ações Externas
Fx = 0.0;
Fy = -g;
Mz = 0.0;

//Definição das condições iniciais

x0 = 0.0;                 // Posição longitudinal inicial da bola
y0 = 2.0;                 // Posição vertical inicial da bola
teta0 = 0.0;              // Posição angular inicial da bola
xpto0 = 3.0;              // Velocidade longitudinal inicial da bola
ypto0 = 6.0;              // Velocidade vertical inicial da bola
tetapto0 = -90.0;         // Velocidade angular inicial da bola

// Formação do vetor de estados z0 (vetor coluna das 6 condições iniciais)

z0(1) = x0;
z0(2) = y0;
z0(3) = teta0;
z0(4) = xpto0;
z0(5) = ypto0;
z0(6) = tetapto0;

// Instantes de tempo para o integrador
t0 = 0.0;                 // Instante de tempo inicial
tf = 8.0;                 // Instante de tempo final
dt = 0.1;                 // Incremento de tempo utilizado na saída do integrador
```

```
//A função zponto contendo a descrição do sistema Az para ser integrado numericamente
// onde z é o vetor de estados e t o instante de tempo

function zpto=bola(t, z)
    zpto(1) = z(4);
    zpto(2) = z(5);
    zpto(3) = z(6);
    zpto(4) = Fx;
    zpto(5) = Fy;
    zpto(6) = Mz;
endfunction

// Função de interrupção associada com opção "roots" da ODE

function w=interrup(t, z)
     w = z(2);        // impacto na posição z(2) = y = 0
endfunction


// Integração numérica da função bola usando a função Ordinary Differential Equation - ODE

zout = [];          // cria vetor de saída
tout  = [];         // cria tempo de saída

while (t0 < tf)     //loop quando houver impacto (interrupção)

   t = t0:dt:tf;              // gera novamente instantes de tempo para integração (de t0 a tf em intervalos dt)

   [z,rd] = ode("roots", z0, t0, t, bola, 1, interrup);    // INTEGRADOR onde z é a saída até interromper no instante
do impacto (rd)

   np = size(z); np = np(2);               // tamanho da saída até o impacto
   t_hit = rd(1);                          // instante do impacto
   zout = [zout z];                        // salva estados de saída até impacto
   tout = [tout [t(1:(np-1)) t_hit]];      // separa trecho de instantes de tempo (-1) + instante impacto
   t0 = rd(1);                             // novo instante de tempo inicial

  //Novas condições iniciais depois do impacto com restituição ey
   z0(1) = z(1,np);
   z0(2) = z(2,np)+100*%eps;
   z0(3) = z(3,np);
   z0(4) = (2*r*z(6)+5*z(4))/7;
   z0(5) = (ey*z(5,np))
   z0(6) = -(2*r*z(6)+5*z(4))/(7*r);

end

tout = [tout tout($)+dt];      // acrescenta o ultimo instante de tempo


// APRESENTAÇÃO DOS RESULTADOS

scf(1),xgrid;
xtitle("POSIÇÃO DA BOLA");
xlabel ('Tempo (s)');
ylabel ('Posição (m)');
plot(tout,zout(1,:));
```

```
scf(2),xgrid;
xtitle("ALTURA DA BOLA");
xlabel ('Tempo (s)');
ylabel ('Altura (m)');
plot(tout,zout(2,:),'o-');

scf(4),xgrid;
xtitle("VELOCIDADE DA BOLA");
xlabel ('Tempo (s)');
ylabel ('Velocidade X (m/s)');
plot(tout,zout(4,:));

scf(5),xgrid;
xtitle("VELOCIDADE DA BOLA");
xlabel ('Tempo (s)');
ylabel ('Velocidade Y (m/s)');
plot(tout,zout(5,:));

scf(6),xgrid;
xtitle("VELOCIDADE ANGULAR DA BOLA");
xlabel ('Tempo (s)');
ylabel ('Velocidade Angular (rad/s)');
plot(tout,zout(6,:));

scf(7),xgrid;
xtitle("MOVIMENTO DA BOLA");
xlabel ('Posição(m)');
ylabel ('Altura (m)');
plot(zout(1,:),zout(2,:),'o-');

scf(8),xgrid;
plot(zout(2,:),zout(5,:));
xtitle("PLANO DE FASE");
xlabel ('Posição (m)');
ylabel ('Velocidade (m/s)');
```